Example 8.8 Diffusion with Reaction
Equation (8.1.18) is solved in Maple below:

```
> restart : with(inttrans) : with(plots) :
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2)-Phi^2*u(x,t);
```

$$eq := \frac{\partial}{\partial t} u(x,t) = \frac{\partial^2}{\partial x^2} u(x,t) - \Phi^2 u(x,t) \quad (1)$$

```
> u(x,0):=0;
```

$$u(x,0) := 0 \quad (2)$$

```
> bc1:=u(x,t)=1;
```

$$bc1 := u(x,t) = 1 \quad (3)$$

```
> bc2:=diff(u(x,t),x)=0;
```

$$bc2 := \frac{\partial}{\partial x} u(x,t) = 0 \quad (4)$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) = \frac{d^2}{dx^2} U(x) - \Phi^2 U(x) \quad (5)$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),bc1);
```

$$bc1 := U(x) = \frac{1}{s} \quad (6)$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),bc2);
```

$$bc2 := \frac{d}{dx} U(x) = 0 \quad (7)$$

```
> dsolve(eqs,U(x));
```

$$U(x) = \_C1 \sin\left(\sqrt{-s - \Phi^2}\, x\right) + \_C2 \cos\left(\sqrt{-s - \Phi^2}\, x\right) \quad (8)$$

```
> U(x):=c[1]*cosh((s+Phi^2)^(1/2)*x)+c[2]*sinh((s+Phi^2)^(1/2)*x);
```

$$U(x) := c_1 \cosh\left(\sqrt{s + \Phi^2}\, x\right) + c_2 \sinh\left(\sqrt{s + \Phi^2}\, x\right) \quad (9)$$

```
> eq0:=eval(subs(x=0,bc1)):
> eq1:=eval(subs(x=1,bc2)):
> con:=solve({eq0,eq1},{c[1],c[2]}):
> U(x):=subs(con,U(x));
```

$$U(x) := \frac{\cosh\left(\sqrt{s + \Phi^2}\, x\right)}{s} - \frac{\sinh\left(\sqrt{s + \Phi^2}\right) \sinh\left(\sqrt{s + \Phi^2}\, x\right)}{\cosh\left(\sqrt{s + \Phi^2}\right) s} \quad (10)$$

```
> U(x):=combine(simplify(U(x))):
```

The shifting theorem is used to find the inverse Laplace transform[!] as:: L-1F(s)=exp(-Φ2t)L=1F(s-Φ2)
.

```
> U(x):=factor(U(x));
```

$$U(x) := \frac{\cosh\left(\sqrt{s+\Phi^2}\,(x-1)\right)}{\cosh\left(\sqrt{s+\Phi^2}\right)s} \tag{11}$$

```
> U1(x):=subs(s=s-Phi^2,U(x));
```

$$U1(x) := \frac{\cosh\left(\sqrt{s}\,(x-1)\right)}{\cosh\left(\sqrt{s}\right)\left(s-\Phi^2\right)} \tag{12}$$

```
> P(s):=numer(U1(x));
```

$$P(s) := -\cosh\left(\sqrt{s}\,(x-1)\right) \tag{13}$$

```
> Q(s):=denom(U1(x));
```

$$Q(s) := \cosh\left(\sqrt{s}\right)\left(-s+\Phi^2\right) \tag{14}$$

```
> A(s):=P(s)/diff(Q(s),s);
```

$$A(s) := -\frac{\cosh\left(\sqrt{s}\,(x-1)\right)}{\frac{1}{2}\,\frac{\sinh\left(\sqrt{s}\right)\left(-s+\Phi^2\right)}{\sqrt{s}}-\cosh\left(\sqrt{s}\right)} \tag{15}$$

```
> solve(Q(s),s);
```

$$-\frac{1}{4}\,\pi^2,\,\Phi^2 \tag{16}$$

```
> _EnvAllSolutions := true:
> solve(Q(s),s):
```

The roots are:

```
> Phi^2,-((2*n-1)*Pi/2)^2;
```

$$\Phi^2,\,-\frac{1}{4}\,(2\,n-1)^2\,\pi^2 \tag{17}$$

```
> A[n]:=simplify(subs(s=mu,A(s)));
```

$$A_n := -\frac{2\cosh\left(\sqrt{\mu}\,(x-1)\right)\sqrt{\mu}}{-\sinh\left(\sqrt{\mu}\right)\mu+\sinh\left(\sqrt{\mu}\right)\Phi^2-2\cosh\left(\sqrt{\mu}\right)\sqrt{\mu}} \tag{18}$$

```
> A[0]:=subs(mu^(1/2)=Phi,mu=Phi^2,A[n]):
> A[0]:=simplify(A[0]);
```

$$A_0 := \frac{\cosh\left(\Phi\,(x-1)\right)}{\cosh(\Phi)} \tag{19}$$

```
> A[n]:=simplify(subs(mu^(1/2)=I*(2*n-1)/2*Pi,mu=-((2*n-1)*Pi/2)
  ^2,A[n])):
```

```
> vars:={cos(1/2*(2*n-1)*Pi)=0,sin(1/2*(2*n-1)*Pi)=(-1)^(n-1)}:
> A[n]:=simplify(subs(vars,A[n]));
```

$$A_n := \frac{4\,(-1)^{-n}\,(2\,n-1)\,\pi\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,(x-1)\right)}{4\,\pi^2\,n^2-4\,\pi^2\,n+\pi^2+4\,\Phi^2} \tag{20}$$

```
> u0s:=A[0]*subs(mu=Phi^2,1/(s-mu));
```

$$u0s := \frac{\cosh(\Phi\,(x-1))}{\cosh(\Phi)\,\left(s-\Phi^2\right)} \tag{21}$$

```
> u0t:=invlaplace(u0s,s,t);
```

$$u0t := \frac{\cosh(\Phi\,(x-1))\,\mathrm{e}^{\Phi^2 t}}{\cosh(\Phi)} \tag{22}$$

```
> uns:=A[n]/(s-mu);
```

$$uns := \frac{4\,(-1)^{-n}\,(2\,n-1)\,\pi\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,(x-1)\right)}{\left(4\,\pi^2\,n^2-4\,\pi^2\,n+\pi^2+4\,\Phi^2\right)\,(s-\mu)} \tag{23}$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := \frac{4\,(-1)^{-n}\,(2\,n-1)\,\pi\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,(x-1)\right)\mathrm{e}^{\mu\,t}}{4\,\pi^2\,n^2-4\,\pi^2\,n+\pi^2+4\,\Phi^2} \tag{24}$$

```
> unt:=subs(mu=-((2*n-1)/2*Pi)^2,unt);
```

$$unt := \frac{4\,(-1)^{-n}\,(2\,n-1)\,\pi\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,(x-1)\right)\mathrm{e}^{-\frac{1}{4}\,(2\,n-1)^2\,\pi^2\,t}}{4\,\pi^2\,n^2-4\,\pi^2\,n+\pi^2+4\,\Phi^2} \tag{25}$$

The time domain solution is obtained by multiplying the inverse Laplace transform of U1(x) by exp(-Φ2t):

```
> U:=simplify(u0t*exp(-Phi^2*t))+Sum(unt,n=1..infinity)*exp(-
  Phi^2*t);
```

$$U := \frac{\cosh(\Phi\,(x-1))}{\cosh(\Phi)} + \left(\sum_{n=1}^{\infty} \frac{4\,(-1)^{-n}\,(2\,n-1)\,\pi\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,(x-1)\right)\mathrm{e}^{-\frac{1}{4}\,(2\,n-1)^2\,\pi^2\,t}}{4\,\pi^2\,n^2-4\,\pi^2\,n+\pi^2+4\,\Phi^2}\right)\mathrm{e}^{-\Phi^2\,t} \tag{26}$$

```
> u:=piecewise(t=0,0,t>0,subs(infinity=20,U)):
```

The following plots are obtained:

```
> plot3d(subs(Phi=1,u),x=1..0,t=0.5..0,axes=boxed,title="Figure
```

```
Exp. 8.15.",labels=[x,t,"u"],orientation=[-45,60]);
```

Figure Exp. 8.15.




```
>
```

The solution obtained matches the separation of variables solution obtained in example 7.8.